

Towards a miniKanren with fair search strategies

KUANG-CHEN LU, Indiana University

WEIXI MA, Indiana University

DANIEL P. FRIEDMAN, Indiana University

We describe fairness levels in disjunction and conjunction implementations. Specifically, a disjunction implementation can be fair, almost-fair, or unfair. And a conjunction implementation can be fair or unfair. We compare the fairness level of four search strategies: the standard miniKanren interleaving depth-first search, the balanced interleaving depth-first search, the fair depth-first search, and the standard breadth-first search. The two non-standard depth-first searches are new. And we present a new, more efficient and shorter implementation of the standard breadth-first search. Using quantitative evaluation, we argue that each depth-first search is a competitive alternative to the standard one, and that our improved breadth-first search implementation is more efficient than the current one.

ACM Reference Format:

Kuang-Chen Lu, Weixi Ma, and Daniel P. Friedman. 2019. Towards a miniKanren with fair search strategies. 1, 1 (May 2019), 15 pages. <https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

1 INTRODUCTION

miniKanren is a family of relational programming languages. Friedman et al. [3, 4] introduce miniKanren and its implementation in *The Reasoned Schemer* and *The Reasoned Schemer, 2nd Ed* (TRS2). After Hemann et al. [5] describe microKanren, a minimal core of miniKanren comprised of only 54 LOC, miniKanren has been implemented in many other languages, including multiple ones using the same language. (e.g. OCanren[7]). Byrd et al. [2] have demonstrated that miniKanren problems that at first reading seem impossible are indeed possible by using a Scheme interpreter written in miniKanren, thereby having access to logic variables within Scheme. miniKanren.org contains links to many miniKanren implementations and the seeds of many interesting problems and their solutions.

A subtlety arises when a cond^e contains many clauses: not every clause has an equal chance of contributing to the result. As an example, consider the following relation repeat^o and its invocation.

^{c1}LKC: submission deadline: Mon 27 May 2019

^{c2}LKC: I have restructured the following paragraph.

^{c3}DPF: I thought we agreed to use BFSser throughout? In that sense it had emerged a few days ago, and now it has re-emerged. It is what we are going to use instead of “current”.

^{c4}LKC: I am confused. Do you mean we are going to use “BFS” for the search strategy, “BFSser” for Seres’s BFS implementation, and “the improved BFS” for our BFS implementation ?

Authors’ addresses: Kuang-Chen LuIndiana University; Weixi MaIndiana University; Daniel P. FriedmanIndiana University.

Unpublished working draft. Not for distribution.

Permission to make digital or hard copies of all or part of this work for personal or classroom use is granted without fee provided that copies are not made or distributed for profit or commercial advantage and that copies bear this notice and the full citation on the first page. Copyrights for components of this work owned by others than ACM must be honored. Abstracting with credit is permitted. To copy otherwise, or republish, to post on servers or to redistribute to lists, requires prior specific permission and/or a fee. Request permissions from permissions@acm.org.

© 2019 Association for Computing Machinery.

XXXX-XXXX/2019/5-ART \$15.00

<https://doi.org/10.1145/nnnnnnnn.nnnnnnnn>

```

48      (defrel (repeato x out)
49      (conde
50      ((≡ `(,x) out))
51      ((fresh (res)
52      (≡ `(,x . ,res) out)
53      (repeato x res))))))
54
55 > (run 4 q
56     (repeato '* q))
57 '((*) (* *) (* * *) (* * * *))

```

Next, consider the following disjunction of invoking repeat^o with four different letters.

```

59 > (run 12 q
60     (conde
61     ((repeato 'a q))
62     ((repeato 'b q))
63     ((repeato 'c q))
64     ((repeato 'd q))))

```

cond^e intuitively relates its clauses with logical or. And thus an unsuspecting beginner would expect each letter to contribute equally to the result, as follows.

```

68 '((a) (b) (c) (d)
69    (a a) (b b) (c c) (d d)
70    (a a a) (b b b) (c c c) (d d d))

```

The cond^e in TRS2, however, generates a less expected result.

```

73 '((a) (a a) (b) (a a a)
74    (a a a a) (b b)
75    (a a a a a) (c)
76    (a a a a a a) (b b b)
77    (a a a a a a a) (d))

```

The miniKanren in TRS2 implements interleaving DFS (DFS_i), the cause of this unexpected result. With this search strategy, each cond^e clause takes half of its received computational resources and passes the other half to its following clauses, except for the last clause that takes all resources it receives. In the example above, the a clause takes half of all resources. And the b clause takes a quarter. Thus c and d barely contribute to the result.

DFS_i is sometimes powerful for an expert. By carefully organizing the order of cond^e clauses, a miniKanren program can explore more “interesting” clauses than those uninteresting ones, and thus use computational resources efficiently.

DFS_i is not always the best choice. For instance, it might be less desirable for novice miniKanren users—understanding implementation details and fiddling with clause order is not their first priority. There is another reason that miniKanren could use more search strategies than just DFS_i. In many applications, there does not exist one order that serves all purposes. For example, a relational dependent type checker contains clauses for constructors that build data and clauses for eliminators that use data. When the type checker is generating simple and shallow programs, the clauses for constructors had better be at the top of the cond^e expression. When performing proof searches for complicated programs, the clauses for eliminators had better be at the top of the cond^e expression. With DFS_i, these two uses cannot be efficient at the same time. In fact, to make one use

efficient, the other one must be more sluggish. Boskin et al. [1] propose and implement a means to eliminate or re-order disjunctive clauses to accommodate varying search needs such as these.

^{c1} The specification that gives every clause in the same cond^e equal “search priority” is fair *disj*. And search strategies with almost-fair *disj* give every clause similar priority. Fair *conj*, a related concept, is more subtle. We cover it in the next section.

To summarize our contributions, we

- propose and implement **balanced interleaving** depth-first search (DFS_{bi}), a new search strategy with almost-fair *disj*.
- propose and implement **fair** depth-first search (DFS_f), a new search strategy with fair *disj*.
- implement in a new way the standard breath-first search (BFS) by Seres et al. [9], a search strategy with fair *disj* and fair *conj*. We refer to “the current BFS” as their implementation and “the improved BFS” as our new one. We formally prove that the two implementations are semantically equivalent, however, the improved BFS runs faster in all benchmarks and is shorter.

2 SEARCH STRATEGIES AND FAIRNESS

In this section, we define fair *disj*, almost-fair *disj* and fair *conj*. Before going further into fairness, we give a short review of the terms: *state*, *space*, and *goal*. A *state* is a collection of constraints. (Here, we restrict constraints to unification constraints.) Every answer corresponds to a state. A *space* is a collection of states. And a *goal* is a function from a state to a space.

Now we elaborate fairness by running more queries about repeat^o . We never use run^* in this paper because fairness is more interesting when we have an unbounded number of answers. It is perfectly fine, however, to use run^* with any search strategy.

2.1 Fair *disj*

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer list. DFS_i , TRS2’s search strategy, however, results in many more lists of *a* than lists of other letters. And some letters (e.g. *c* and *d*) are rarely seen. The more clauses, the worse the situation.

```
;;  $\text{DFS}_i$  (unfair disj)
> (run 12 q
  (conde
    ((repeato 'a q))
    ((repeato 'b q))
    ((repeato 'c q))
    ((repeato 'd q))))
'((a) (a a) (b) (a a a)
  (a a a a) (b b)
  (a a a a a) (c)
  (a a a a a a) (b b b)
  (a a a a a a a) (d))
```

^{c1} LKC: I have cited Boskin in the above paragraph.

Under the hood, the `conde` here allocates computational resources to four trivially different spaces. The unfair `disj` in `DFSi` allocates many more resources to the first space. On the contrary, fair `disj` would allocate resources evenly to each space.

<pre> 142 ;; DFS_f (fair disj) 143 > (run 12 q 144 (cond^e 145 ((repeat^o 'a q)) 146 ((repeat^o 'b q)) 147 ((repeat^o 'c q)) 148 ((repeat^o 'd q)))) 149 '((a) (b) (c) (d) 150 (a a) (b b) (c c) (d d) 151 (a a a) (b b b) (c c c) (d d d)) </pre>	<pre> 142 ;; BFS (fair disj) 143 > (run 12 q 144 (cond^e 145 ((repeat^o 'a q)) 146 ((repeat^o 'b q)) 147 ((repeat^o 'c q)) 148 ((repeat^o 'd q)))) 149 '((a) (b) (c) (d) 150 (a a) (b b) (c c) (d d) 151 (a a a) (b b b) (c c c) (d d d)) </pre>
---	---

Running the same program again with almost-fair `disj` (e.g. `DFSbi`) gives the same result. Almost-fair, however, is not completely fair, as shown by the following example.

```

159 ;; DFSbi (almost-fair disj)
160 > (run 16 q
161   (conde
162     ((repeato 'a q))
163     ((repeato 'b q))
164     ((repeato 'c q))
165     ((repeato 'd q))
166     ((repeato 'e q))))
167 '((b) (c) (d) (a)
168   (b b) (c c) (d d) (e)
169   (b b b) (c c c) (d d d) (a a)
170   (b b b b) (c c c c) (d d d d) (e e))

```

`DFSbi` is fair only when the number of goals is a power of 2, otherwise, it allocates some goals with twice as many resources as the others. In the above example, where the `conde` has five clauses, `DFSbi` allocates more resources to the clauses of b, c, and d.

We end this subsection with precise definitions of all levels of `disj` fairness. Our definition of *fair disj* is slightly more general than the one in Seres et al. [9], which is only for binary disjunction. We generalize it to a multi-arity one.

DEFINITION 2.1 (FAIR `disj`). *A `disj` is fair if and only if it allocates computational resources evenly to spaces produced by goals in the same disjunction (i.e., clauses in the same `conde`).*

DEFINITION 2.2 (ALMOST-FAIR `disj`). *A `disj` is almost-fair if and only if it allocates computational resources so evenly to spaces produced by goals in the same disjunction that the maximal ratio of resources is bounded by a constant.*

DEFINITION 2.3 (UNFAIR `disj`). *A `disj` is unfair if and only if it is not almost-fair.*

2.2 Fair conj

Given the following program, it is natural to expect lists of each letter to constitute 1/4 in the answer list. Search strategies with unfair conj: DFS_i , DFS_{bi} , and DFS_f , however, results in many more lists of a than lists of other letters. And some letters are rarely seen. Here again, as the number of clauses grows, the situation worsens.

Although some strategies have a different level of fairness in disj, they have the same behavior when there is no call to a relational definition in cond^e clauses (including this case).

<pre> ;; DFS_i (unfair conj) > (run 12 q (fresh (x) (cond^e ((= 'a x)) ((= 'b x)) ((= 'c x)) ((= 'd x))) (repeat^o x q))) '((a) (a a) (b) (a a a) (a a a a) (b b) (a a a a a) (c) (a a a a a a) (b b b) (a a a a a a a) (d)) </pre>	<pre> ;; DFS_f (unfair conj) > (run 12 q (fresh (x) (cond^e ((= 'a x)) ((= 'b x)) ((= 'c x)) ((= 'd x))) (repeat^o x q))) '((a) (a a) (b) (a a a) (a a a a) (b b) (a a a a a) (c) (a a a a a a) (b b b) (a a a a a a a) (d)) </pre>	<pre> ;; DFS_{bi} (unfair conj) > (run 12 q (fresh (x) (cond^e ((= 'a x)) ((= 'b x)) ((= 'c x)) ((= 'd x))) (repeat^o x q))) '((a) (a a) (c) (a a a) (a a a a) (c c) (a a a a a) (b) (a a a a a a) (c c c) (a a a a a a a) (d)) </pre>
---	---	--

Under the hood, the cond^e and the call to repeat^o are connected by conj. The cond^e goal outputs a space including four trivially different states. Applying the next conjunctive goal, $(\text{repeat}^o x q)$, produces four trivially different spaces. In the examples above, all search strategies allocate more computational resources to the space of a. On the contrary, fair conj would allocate resources evenly to each space. For example,

```

;; BFS (fair conj)
> (run 12 q
   (fresh (x)
    (conde
      ((= 'a x))
      ((= 'b x))
      ((= 'c x))
      ((= 'd x)))
    (repeato x q)))
'((a) (b) (c) (d)
  (a a) (b b) (c c) (d d)
  (a a a) (b b b) (c c c) (d d d))

```

A more interesting situation is when the first conjunct produces an unbounded number of states. Consider the following example: a naive specification of fair conj might require search strategies to produce all sorts of singleton lists, but there would not be any lists of length two or longer, which makes the strategies incomplete. A search strategy is *complete* if and only if “every correct answer would be discovered after some finite time” [9], otherwise, it is *incomplete*. In the context of miniKanren, a search strategy is complete means that every correct answer has a position in large enough answer lists.

```

236 ;; naively fair conj
237 > (run 6 q
238   (fresh (xs)
239     (conde
240       ((repeato 'a xs))
241       ((repeato 'b xs)))
242     (repeato xs q)))
243 '(((a)) ((b))
244   ((a a)) ((b b))
245   ((a a a)) ((b b b)))

```

Our solution requires a search strategy with *fair conj* to organize states in buckets in spaces, where each bucket contains finite states, and to allocate resources evenly among spaces derived from states in the same bucket. It is up to a search strategy designer to decide by what criteria to put states in the same bucket, and how to allocate resources among spaces related to different buckets.

BFS puts states of the same cost in the same bucket, and allocates resources carefully among spaces related to different buckets such that it produces answers in increasing order of cost. The *cost* of an answer is its depth in the search tree (i.e., the number of calls to relational definitions required to find the answer) [9]. In the above examples, the cost of each answer is equal to their lengths because we need to apply *repeat^o* n times to find an answer of length n . In the following example, every answer is a list of a list of symbols, where inner lists in the same outer list are identical. Here the cost of each answer is equal to the length of its inner lists plus the length of its outer list. For example, the cost of $((a) (a))$ is $1 + 2 = 3$.

```

259 ;; BFS (fair conj)
260 > (run 12 q
261   (fresh (xs)
262     (conde
263       ((repeato 'a xs))
264       ((repeato 'b xs)))
265     (repeato xs q)))
266 '(((a)) ((b))
267   ((a) (a)) ((b) (b))
268   ((a a)) ((b b))
269   ((a) (a) (a)) ((b) (b) (b))
270   ((a a) (a a)) ((b b) (b b))
271   ((a a a)) ((b b b)))

```

We end this subsection with precise definitions of all levels of *conj* fairness.

DEFINITION 2.4 (FAIR conj). *A conj is fair if and only if it allocates computational resources evenly to spaces produced from states in the same bucket. A bucket is a finite collection of states. And search strategies with fair conj should represent spaces with possibly unbounded collections of buckets.*

DEFINITION 2.5 (UNFAIR conj). *A conj is unfair if and only if it is not fair.*

```

283 #| Goal × Goal → Goal |#
284 (define (disj2 g1 g2)
285   (lambda (s)
286     (append∞ (g1 s) (g2 s))))
287
288 #| Space × Space → Space |#
289 (define (append∞ s∞ t∞)
290   (cond
291     ((null? s∞) t∞)
292     ((pair? s∞)
293      (cons (car s∞)
294            (append∞ (cdr s∞) t∞)))
295     (else (lambda ()
296              (append∞ t∞ (s∞))))))
297
298
299 (define-syntax disj
300   (syntax-rules ()
301     ((disj) (fail))
302     ((disj g0 g ...) (disj+ g0 g ...))))
303
304
305 (define-syntax disj+
306   (syntax-rules ()
307     ((disj+ g) g)
308     ((disj+ g0 g1 g ...) (disj2 g0 (disj+ g1 g ...))))
309
310

```

Fig. 1. implementation of DFS_i (Part I)

3 INTERLEAVING DEPTH-FIRST SEARCH

In this section, we review the implementation of interleaving depth-first search (DFS_i). We focus on parts that are relevant to this paper. TRS2, chapter 10 and the appendix, “Connecting the wires”, provide a comprehensive description of the miniKanren implementation but limited to unification constraints (\equiv). Fig. 1 and Fig. 2 show parts that are later compared with other search strategies. We follow some conventions to name variables: ss name states; gs (possibly with subscript) name goals; variables ending with $^\infty$ name (search) spaces. Fig. 1 shows the implementation of $disj$. The first function, $disj_2$, implements binary disjunction. It applies the two disjunctive goals to the input state s and composes the two resulting spaces with $append^\infty$. The following syntax definitions say $disj$ is right-associative. Fig. 2 shows the implementation of $conj$. The first function, $conj_2$, implements binary conjunction. It applies the *first* goal to the input state, then applies the second goal to states in the resulting space. The helper function $append-map^\infty$ applies its input goal to states in its input space and composes the resulting spaces. It reuses $append^\infty$ for space composition. The following syntax definitions say $conj$ is also right-associative.

```

330 #| Goal × Goal → Goal |#
331 (define (conj2 g1 g2)
332   (lambda (s)
333     (append-map∞ g2 (g1 s))))
334
335 #| Goal × Space → Space |#
336 (define (append-map∞ g s∞)
337   (cond
338     ((null? s∞) '())
339     ((pair? s∞)
340      (append∞ (g (car s∞))
341                (append-map∞ g (cdr s∞))))
342     (else (lambda ()
343              (append-map∞ g (s∞))))))
344
345 (define-syntax conj
346   (syntax-rules ()
347     ((conj) (fail))
348     ((conj g0 g ...) (conj+ g0 g ...))))
349
350 (define-syntax conj+
351   (syntax-rules ()
352     ((conj+ g) g)
353     ((conj+ g0 g1 g ...) (conj2 g0 (conj+ g1 g ...))))
354
355
356
357
358
359
360

```

Fig. 2. implementation of DFS_i (Part II)

4 BALANCED INTERLEAVING DEPTH-FIRST SEARCH

The strategy **balanced interleaving DFS** (DFS_{bi}) has an almost-fair disj and unfair conj. The implementation of DFS_{bi} differs from DFS_i's in the disj macro. When there are one or more disjunctive goals, the new disj builds a balanced binary tree whose leaves are the goals and whose nodes are disj₂s, hence the name of this search strategy. In contrast, the disj in DFS_i constructs the binary tree in a particularly unbalanced form. We list the new disj with its helper in Fig. 3. The new helper, disj+, takes two additional 'arguments'. They accumulate goals to be put in the left and right subtrees. The first clause handles the case where there is only one goal. In this case, the tree is the goal itself. When there are more goals, we partition the list of goals into two sublists of roughly equal lengths and recur on the two sublists. We move goals to the accumulators in the last clause. As we are moving two goals each time, there are two base cases: (1) no goal remains; (2) one goal remains. We handle these two new base cases in the second clause and the third clause, respectively.

5 FAIR DEPTH-FIRST SEARCH

Fair DFS (DFS_f) has fair disj and unfair conj. The implementation of DFS_f differs from DFS_i's in disj₂ (Fig. 4). The new disj₂ calls a new and fair version of append[∞]. append[∞]_{fair} immediately calls its helper, loop, with the


```

377 (define-syntax disj
378   (syntax-rules ()
379     ((disj) fail)
380     ((disj g ...) (disj+ (g ...) () ())))))
381
382 (define-syntax disj+
383   (syntax-rules ()
384     ((disj+ () () g) g)
385     ((disj+ (gl ...) (gr ...))
386      (disj2 (disj+ () () gl ...)
387              (disj+ () () gr ...)))
389     ((disj+ (gl ...) (gr ...) g0)
390      (disj2 (disj+ () () gl ... g0)
391              (disj+ () () gr ...)))
392     ((disj+ (gl ...) (gr ...) g0 g1 g ...)
393      (disj+ (gl ... g0) (gr ... g1) g ...)))
394
395
396
397
398
399
400
401
402
403
404
405
406
407
408
409
410
411
412
413
414
415
416
417
418
419
420
421
422
423

```

Fig. 3. implementation of DFS_{bi}

```

398 #| Goal × Goal → Goal |#
399 (define (disj2 g1 g2)
400   (lambda (s)
401     (append∞fair (g1 s) (g2 s))))
402
403
404 #| Space × Space → Space |#
405 (define (append∞fair s∞ t∞)
406   (let loop ((s? #t) (s∞ s∞) (t∞ t∞))
407     (cond
408       ((null? s∞) t∞)
409       ((pair? s∞)
410        (cons (car s∞)
411              (loop s? (cdr s∞) t∞)))
412       (s? (loop #f t∞ s∞))
413       (else (lambda ()
414                (loop #t (t∞) (s∞)))))))
415
416
417
418
419
420
421
422
423

```

Fig. 4. implementation of DFS_f

first argument, $s?$, initialized to $\#t$, which indicates that we haven't swapped s^∞ and t^∞ . The swapping happens at the third `cond` clause in `loop`, where $s?$ is updated accordingly. The first two `cond` clauses essentially copy the cars and stop recursion when one of the input spaces is obviously finite. The third clause, as we mentioned

```

424 #| Goal × Space → Space |#
425 (define (append-map∞fair g s∞)
426   (cond
427     ((null? s∞) '())
428     ((pair? s∞)
429      (append∞fair (g (car s∞))
430                    (append-map∞fair g (cdr s∞))))
431     (else (lambda ()
432              (append-map∞fair g (s∞)))))))
433
434
435

```

Fig. 5. stepping-stone toward the improved BFS (based on DFS_f)

above, is just for swapping. When the fourth and last clause runs, we know that both s^{∞} and t^{∞} end with thunks, and that we have swapped them. In this case, we construct a new thunk. The new thunk swaps back the two spaces in the recursive call to loop. This is unnecessary for fairness—we do it to produce answers in a more readable order.

c1

6 BREADTH-FIRST SEARCH

BFS has both fair disj and fair conj. Our first BFS implementation (Fig. 5) serves as a “stepping-stone” toward the improved BFS. It is so similar to DFS_f (not DFS_i) that we only need to apply two changes: (1) rename $\text{append-map}^{\infty}$ to $\text{append-map}^{\infty}_{\text{fair}}$ and (2) replace append^{∞} with $\text{append}^{\infty}_{\text{fair}}$ in $\text{append-map}^{\infty}_{\text{fair}}$ ’s body.

This implementation can be improved in two ways. First, as mentioned in subsection 2.2, BFS puts answers in buckets and answers of the same cost are in the same bucket. In the above implementation, however, it is not obvious how we manage cost information—the cars of a space have cost 0 (i.e., they are all in the same bucket), and every thunk indicates an increment in cost. It is even more subtle that $\text{append}^{\infty}_{\text{fair}}$ and the $\text{append-map}^{\infty}_{\text{fair}}$ respects the cost information. Second, $\text{append}^{\infty}_{\text{fair}}$ is extravagant in memory usage. It makes $O(n + m)$ new cons cells every time, where n and m are the sizes of the first buckets of two input spaces. DFS_f is also space extravagant.

In the following subsections, we first describe the improved BFS implementation that manages cost information in a more clear and concise way and is less extravagant in memory usage. Then we compare the improved BFS with the current BFS.

6.1 Improved BFS

We simplify the cost information by changing the Space type, modifying related function definitions, and introducing a few more functions.

The new type of Space is a pair whose car is a list of answers (the bucket), and whose cdr is either #f or a thunk returning a space. The #f here means the space is obviously finite, just like empty list in other implementations.

c2

^{c1}LKC: Following Boskin’s suggestion, “are ended with a thunk” -> “end with thunks”

^{c2}LKC: I have rewritten the sentence about #f.

```

471 #|  $\rightarrow \text{Space}$  |#
472 (define (none)
473   `(() . #f))
474
475 #|  $\text{State} \rightarrow \text{Space}$  |#
476 (define (unit s)
477   `((,s) . #f))
478
479
480 #|  $(\rightarrow \text{Space}) \rightarrow \text{Space}$  |#
481 (define (step f)
482   `(() . ,f))
483
484 #|  $\text{Space} \times \text{Space} \rightarrow \text{Space}$  |#
485 (define (append∞fair s∞ t∞)
486   (cons (append (car s∞) (car t∞))
487     (let ((t1 (cdr s∞)) (t2 (cdr t∞)))
488       (cond
489         ((not t1) t2)
490         ((not t2) t1)
491         (else (lambda ()
492                   (append∞fair (t1) (t2)))))))
493
494
495 #|  $\text{Goal} \times \text{Space} \rightarrow \text{Space}$  |#
496 (define (append-map∞fair g s∞)
497   (foldr
498     (lambda (s t∞)
499       (append∞fair (g s) t∞))
500     (let ((f (cdr s∞)))
501       (step (and f (lambda () (append-map∞fair g (f))))))
502     (car s∞))
503
504
505 #|  $\text{Maybe Nat} \times \text{Space} \rightarrow [\text{State}]$  |#
506 (define (take∞ n s∞)
507   (let loop ((n n) (vs (car s∞)))
508     (cond
509       ((and n (zero? n)) '())
510       ((pair? vs)
511        (cons (car vs)
512              (loop (and n (sub1 n)) (cdr vs))))
513       (else (let ((f (cdr s∞)))
514                (if f (take∞ n (f)) '()))))))
515
516
517

```

Fig. 6. New and changed functions in the improved BFS that implements pure features

```

518 #| Space × (State × Space → Space) × (→ Space) → Space |#
519 (define (elim s∞ fk sk)
520   (let ((ss (car s∞)) (f (cdr s∞)))
521     (cond
522       ((pair? ss) (sk (car ss) (cons (cdr ss) f)))
523       (f (step (lambda () (elim (f) fk sk))))
524       (else (fk)))))
525
526
527 #| Goal × Goal × Goal → Goal |#
528 (define (ifte g1 g2 g3)
529   (lambda (s)
530     (elim (g1 s)
531           (lambda () (g3 s))
532           (lambda (s0 s∞)
533             (append-map∞fair g2
534               (append∞fair (unit s0) s∞))))))
535
536
537 #| Goal → Goal |#
538 (define (once g)
539   (lambda (s)
540     (elim (g s)
541           (lambda () (none))
542           (lambda (s0 s∞) (unit s0))))))
543
544

```

Fig. 7. New and changed functions in the improved BFS that implement impure features

We list functions related to the pure subset in Fig. 6. The first three functions are space constructors. `none` makes an empty space; `unit` makes a space from one answer; and `step` makes a space from a thunk. The remaining functions are as before.

Luckily, the change in `append∞fair` also fixes the miserable space extravagance—the use of `append` helps us to reuse the first bucket of `t∞`.

Kiselyov et al. [6] have demonstrated that a *MonadPlus* hides in implementations of logic programming systems. the improved BFS is not an exception: `append-map∞fair` is like `bind`, but takes arguments in reversed order; `none`, `unit`, and `append∞fair` correspond to `mzero`, `unit`, and `mplus`, respectively.

Functions implementing impure features are in Fig. 7. The first function, `elim`, takes a space `s∞` and two continuations `fk` and `sk`. When `s∞` contains no answers, it calls `fk`. Otherwise, it calls `sk` with the first answer and the rest of the space. This function is similar to an eliminator of spaces, hence the name. The remaining functions are as before.

6.2 Compare the improved BFS with the current BFS

In this subsection, we compare the pure subset of the improved BFS with the current BFS. We focus on the pure subset because the current BFS is designed for a pure relational programming system. We prove in Coq

benchmark	size	DFS _i	DFS _{bi}	DFS _f	improved BFS	current BFS
very-recursive ^o	100000	166	103	412	451	1024
	200000	283	146	765	839	1875
	300000	429	346	2085	1809	4408
append ^o	100	17	18	17	17	65
	200	145	137	137	142	121
	300	388	384	387	429	371
revers ^o	10	3	4	3	18	36
	20	35	35	33	3070	3695
	30	329	333	315	75079	107531
quine-1	1	13	14	10	-	-
	2	30	34	25	-	-
	3	41	48	33	-	-
quine-2	1	22	21	12	-	-
	2	51	46	24	-	-
	3	72	76	32	-	-
'(I love you)-1	99	10	37	-	40	96
	198	22	71	-	374	652
	297	24	292	-	1668	3759
'(I love you)-2	99	485	179	-	44	94
	198	808	638	-	414	630
	297	1265	916	-	1651	3691

Table 1. The results of a quantitative evaluation: running times of benchmarks in milliseconds

that these two search strategies are semantically equivalent, since the result of $(\text{run } n \text{ ? } g)$ is the same either way. (See supplements for the formal proof.) To compare efficiency, we translate the current BFS's Haskell code into Racket (See supplements for the translated code). The translation is direct due to the similarity of the two relational programming systems. The translated code is longer than the improved BFS. And it runs slower in all benchmarks. Details about differences in efficiency are in section 7.

7 QUANTITATIVE EVALUATION

In this section, we compare the efficiency of the search strategies. A concise description is in Table 1. A hyphen means “running out of 500 MB memory”. The first two benchmarks are from TRS2. `reverso` is from Rozplokh and Boulytchev [8]. The next two benchmarks about generating quines are based on a similar test case in Byrd et al. [2]. We modify the relational interpreters because we don't have disequality constraints (e.g. `absento`). The sibling benchmarks differ in the `conde` clause order of their relational interpreters. The last two benchmarks are about synthesizing expressions that evaluate to `'(I love you)`. They are also based on a similar test case in Byrd et al. [2]. Again, we modify the relational interpreters for the same reason. And the sibling benchmarks differ in the `conde` clause order of their relational interpreters. The first one has elimination rules (i.e. `application`, `car`, and `cdr`) at the end, while the other has them at the beginning. We conjecture that DFS_i would perform

badly in the second case because elimination rules complicate the problem when synthesizing (i.e., our evaluation supports our conjecture.)

In general, only DFS_i and DFS_{bi} consistently perform well. DFS_f is just as efficient in `appendo`, `reverso`, and quines. All BFS implementations have obvious overhead in most cases. Among the three variants of DFS, which all have unfair `conj`, DFS_f is most resistant to clause permutation in quines, followed by DFS_{bi} then DFS_i . DFS_f , however, runs out of memory in `'(I love you)s`. In contrast, DFS_{bi} is still more stable than DFS_i . Thus, we consider DFS_{bi} a competitive alternative to DFS_i . Among the two implementations of BFS, the improved BFS constantly performs as well or better. It is hard to conclude how `disj` fairness affects performance based on these statistics. Fair `conj`, however, imposes considerable overhead constantly except in `appendo`. The reason might be that those strategies tend to keep more intermediate answers in the memory.

8 RELATED WORKS

Yang [10] points out that a disjunct complex would be ‘fair’ if it were a full and balanced tree.

Seres et al. [9] describe a breadth-first search strategy. We present another implementation. Our implementation is semantically equivalent to theirs. But, ours is shorter and performs better in comparison with a straightforward translation of their Haskell code.

Rozplokhas and Boulytchev [8] address the non-commutativity of conjunction, while our work about `disj` fairness addresses the non-commutativity of disjunction.

9 CONCLUSION

We analyze the definitions of fair `disj` and fair `conj`, then propose a new definition of fair `conj`. Our definition of fair `conj` is orthogonal with completeness.

We devise two new search strategies (i.e., balanced interleaving DFS (DFS_{bi}) and fair DFS (DFS_f)) and devise a new implementation of BFS. These strategies have different features in fairness: DFS_{bi} has an almost-fair `disj` and unfair `conj`. DFS_f has fair `disj` and unfair `conj`. BFS has both fair `disj` and fair `conj`.

Our quantitative evaluation shows that DFS_{bi} is a competitive alternative to DFS_i , the current miniKanren search strategy, and that the improved BFS is more practical than the current BFS.

We prove formally that the improved BFS is semantically equivalent to the current BFS. But, the improved BFS is shorter and performs better in comparison with a straightforward translation of their Haskell code.

We acknowledge that we are making large claims about efficiency given that we have just a few benchmarks, but our intuition is based on sound principles. ...

ACKNOWLEDGMENTS

REFERENCES

- [1] Benjamin Strahan Boskin, Weixi Ma, David Thrane Christiansen, and Daniel P. Friedman. 2018. A Surprisingly Competetive Conditional Operator: for miniKanrenizing the Inference Rules of Pie (*Scheme '18*). St Louis, MO, USA.
- [2] William E Byrd, Michael Ballantyne, Gregory Rosenblatt, and Matthew Might. 2017. A unified approach to solving seven programming problems (functional pearl). *Proceedings of the ACM on Programming Languages* 1, ICFP (2017).
- [3] Daniel P. Friedman, William E. Byrd, and Oleg Kiselyov. 2005. *The Reasoned Schemer*. The MIT Press.
- [4] Daniel P. Friedman, William E. Byrd, Oleg Kiselyov, and Jason Hemann. 2018. *The Reasoned Schemer, Second Edition*.

^{c1} LKC: (following Boskin’s suggestion) constantly -> consistently

^{c1} DPF: Finish this paragraph with a clarification and you may place this anywhere in the conclusion. This may require some consultation with Weixi.

^{c2} LKC: I have sent an email to Weixi on the clarification.

- [5] Jason Hemann, Daniel P. Friedman, William E. Byrd, and Matthew Might. 2016. A small embedding of logic programming with a simple complete search. In *Proceedings of the 12th Symposium on Dynamic Languages - DLS 2016*. ACM Press. <https://doi.org/10.1145/2989225.2989230>
- [6] Oleg Kiselyov, Chung-chieh Shan, Daniel P. Friedman, and Amr Sabry. 2005. Backtracking, interleaving, and terminating monad transformers:(functional pearl). *ACM SIGPLAN Notices* 40, 9 (2005), 192–203.
- [7] Dmitry Kosarev and Dmitry Boulytchev. 2018. Typed embedding of a relational language in OCaml. *arXiv preprint arXiv:1805.11006* (2018).
- [8] Dmitri Rozplokhas and Dmitri Boulytchev. 2018. Improving Refutational Completeness of Relational Search via Divergence Test. In *Proceedings of the 20th International Symposium on Principles and Practice of Declarative Programming*. ACM, 18.
- [9] Silvija Seres, J. Michael Spivey, and C. A. R. Hoare. 1999. Algebra of Logic Programming.. In *ICLP*. 184–199.
- [10] Edward Z. Yang. 2010. Adventures in Three Monads. *The Monad. Reader Issue* 15 (2010), 11.